
Count unique numeric values with criteria

Generic formula 

{ = SUM( -- (FREQUENCY(IF(criteria,values),values) > 0))}

Summary 

To count unique numeric values in a range, you can use a formula based on the FREQUENCY,
SUM, and IF functions. In the example shown, employee numbers appear in the range B5:B14.
The formula in G6 is:

= SUM( -- (FREQUENCY(IF(C5:C14 = "A",B5:B14),B5:B14) > 0))

which returns 2, since there are 2 unique employee ids in building A.

Note: this is an array formula and must be entered with control + shift + enter, unless you are using
Excel 365.

With Excel 365, you can use a simpler and faster formula based on UNIQUE.

Explanation 

Note: Prior to Excel 365, Excel did not have a dedicated function to count unique values. This formula
shows a one way to count unique values, as long as they are numeric. If you have text values, or a mix
of text and numbers, you'll need to use a more complicated formula.

The Excel FREQUENCY function returns a frequency distribution, which is a summary table that
contains the frequency of numeric values, organized in "bins". We use it here as a roundabout way
to count unique numeric values. To apply criteria, we use the IF function.

Working from the inside-out, we first filter values with the IF function:

IF(C5:C14 = "A",B5:B14) // filter on building A

The result of this operation is an array like this:

{905;905;905;905;773;773;FALSE;FALSE;FALSE;FALSE}

Notice all ids in building B are now FALSE. This array is delivered directly to the FREQUENCY
function as the data_array. For the bins_array, we supply the ids themselves:

FREQUENCY({905;905;905;905;773;773;FALSE;FALSE;FALSE;FALSE},
{905;905;905;905;773;773;801;963;963;963})

With this configuration, FREQUENCY returns the array below:

{4;0;0;0;2;0;0;0;0;0;0}

The result is a bit cryptic, but the meaning is 905 appears four times, and 773 appears two times.
The FALSE values are automatically ignored.

FREQUENCY has a special feature that automatically returns zero for any numbers that have
already appeared in the data array, which is why values are zero once a number has been
encountered. This is the feature that allows this approach to work. 

Next, each of these values is tested to be greater than zero:

{4;0;0;0;2;0;0;0;0;0;0} > 0

The result is an array like this:

{TRUE;FALSE;FALSE;FALSE;TRUE;FALSE;FALSE;FALSE;FALSE;FALSE;FALSE}

Each TRUE in the list represents a unique number in the list, and we just need to add up the TRUE
values with SUM. However, SUM won't add up logical values in an array, so we need to first coerce
the values into 1 or zero. This is done with the double-negative (--). The result an array of only 1's
or 0's:

{1;0;0;0;1;0;0;0;0;0;0}

Finally, SUM adds these values up and returns the total, which in this case is 2.

Multiple criteria

You can extend the formula to handle multiple criteria like this:

{ = SUM( -- (FREQUENCY(IF((criteria1) *
(criteria2),values),values) > 0))}

UNIQUE function in Excel 365

In Excel 365, the UNIQUE function provides a better, more elegant way to list unique values and
count unique values. These formulas can be adapted to apply logical criteria.

Author 

Dave Bruns

Related formulas 

Count unique text values in a range
This formula is more complicated than a similar formula that
uses FREQUENCY to count unique numeric values because
FREQUENCY doesn't work with non-numeric values. As a result, a
large part of the formula simply transforms the non-numeric
data into...

Count unique values in a range with COUNTIF
Working from the inside out, COUNTIF is configured to values in
the range B5:B14, using all of these same values as criteria:
COUNTIF ( B5:B14 , B5:B14 ) Because we provide 10 values for
criteria, we get back an array with 10 results like this: { 3...

Count unique text values with criteria
This is a complex formula that uses FREQUENCY to count
numeric values that are derived with the MATCH function.
Working from the inside out, the MATCH function is used to get
the position of each value that appears in the data: MATCH (
B5:B11 , B5:...

Extract unique items from a list
The core of this formula is a basic lookup with INDEX: = INDEX (
list , row ) In other words, give INDEX the list and a row number,
and INDEX will retrieve a value to add to the unique list. The hard
work is figuring out the ROW number to give INDEX...

Related functions 

Excel FREQUENCY Function
The Excel FREQUENCY function returns a frequency distribution,
which is a list that shows the frequency of values at given
intervals. FREQUENCY returns multiple values and must be
entered as an array formula...

Excel SUM Function
The Excel SUM function returns the sum of values supplied.
These values can be numbers, cell references, ranges, arrays, and
constants, in any combination. SUM can handle up to 255
individual arguments.

Good links 

Mike Girvin's book Control-Shift-Enter

Related courses 

Core Formula

Excel Formula Training
Formulas are the key to getting things done in
Excel. In this accelerated training, you'll learn how to
use formulas to manipulate text, work with dates and
times, lookup values with VLOOKUP and INDEX &
MATCH, count and sum with criteria, dynamically
rank values, and create dynamic ranges. You'll also
learn how to troubleshoot, trace errors, and fix
problems. Instant access. See details here.

500 Formulas | 101 Functions

Download 100+ Important Excel Functions

Get over 100 Excel Functions you should know in one handy PDF.

email address Get the PDF

Topics

Formula Basics
Formula Examples
Conditional Formatting
Pivot Tables
Excel Tables
Excel Charts
Shortcuts
More...

Key functions

IF function
VLOOKUP function
XLOOKUP function
FILTER function
SUMIFS function
COUNTIFS function
SUMPRODUCT function
INDEX and MATCH
More functions...

Hi - I'm Dave Bruns, and
I run Exceljet with my
wife, Lisa. Our goal is to
help you work faster in

Excel. We create short videos, and clear
examples of formulas, functions, pivot
tables, conditional formatting, and
charts. Read more.

Excel video training

Learn more

I am relatively new to Excel and am happy that I discovered your website. I used to go to Youtube
for questions, but your website is much more helpful. -Torsten

